

## F07TJF (STRTRI/DTRTRI) – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

F07TJF (STRTRI/DTRTRI) computes the inverse of a real triangular matrix.

### 2 Specification

```
SUBROUTINE F07TJF(UPLO, DIAG, N, A, LDA, INFO)
ENTRY      strtri(UPLO, DIAG, N, A, LDA, INFO)
INTEGER    N, LDA, INFO
real      A(LDA,*)
CHARACTER*1 UPLO, DIAG
```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

This routine forms the inverse of a real triangular matrix  $A$ . Note that the inverse of an upper (lower) triangular matrix is also upper (lower) triangular.

### 4 References

- [1] Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

### 5 Parameters

- 1:** UPLO — CHARACTER\*1 *Input*  
*On entry:* indicates whether  $A$  is upper or lower triangular as follows:  
     if UPLO = 'U', then  $A$  is upper triangular;  
     if UPLO = 'L', then  $A$  is lower triangular.  
*Constraint:* UPLO = 'U' or 'L'.
- 2:** DIAG — CHARACTER\*1 *Input*  
*On entry:* indicates whether  $A$  is a non-unit or unit triangular matrix as follows:  
     if DIAG = 'N', then  $A$  is a non-unit triangular matrix;  
     if DIAG = 'U', then  $A$  is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.  
*Constraint:* DIAG = 'N' or 'U'.
- 3:** N — INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .

**4:** A(LDA,\*) — *real* array *Input/Output*

**Note:** the second dimension of the array A must be at least  $\max(1,N)$ .

*On entry:* the  $n$  by  $n$  triangular matrix  $A$ . If UPLO = 'U',  $A$  is upper triangular and the elements of the array below the diagonal are not referenced; if UPLO = 'L',  $A$  is lower triangular and the elements of the array above the diagonal are not referenced. If DIAG = 'U', the diagonal elements of  $A$  are not referenced, but are assumed to be 1.

*On exit:*  $A$  is overwritten by  $A^{-1}$ , using the same storage format as described above.

**5:** LDA — INTEGER *Input*

*On entry:* the first dimension of the array A as declared in the (sub)program from which F07TJF (STRTRI/DTRTRI) is called.

*Constraint:*  $LDA \geq \max(1,N)$ .

**6:** INFO — INTEGER *Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO =  $i$ ,  $a_{ii}$  is zero and the matrix  $A$  is singular.

## 7 Accuracy

The computed inverse  $X$  satisfies

$$|XA - I| \leq c(n)\epsilon|X||A|,$$

where  $c(n)$  is a modest linear function of  $n$ , and  $\epsilon$  is the *machine precision*.

Note that a similar bound for  $|AX - I|$  cannot be guaranteed, although it is almost always satisfied.

The computed inverse satisfies the forward error bound

$$|X - A^{-1}| \leq c(n)\epsilon|A^{-1}||A||X|.$$

See Du Croz and Higham [1].

## 8 Further Comments

The total number of floating-point operations is approximately  $\frac{1}{3}n^3$ .

The complex analogue of this routine is F07TWF (CTRTRI/ZTRTRI).

## 9 Example

To compute the inverse of the matrix  $A$ , where

$$A = \begin{pmatrix} 4.30 & 0.00 & 0.00 & 0.00 \\ -3.96 & -4.87 & 0.00 & 0.00 \\ 0.40 & 0.31 & -8.02 & 0.00 \\ -0.27 & 0.07 & -5.95 & 0.12 \end{pmatrix}.$$

## 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F07TJF Example Program Text
*      Mark 15 Release. MAG Copyright 1991.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX, LDA
PARAMETER       (NMAX=8,LDA=NMAX)
CHARACTER       DIAG
PARAMETER       (DIAG='N')
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N
CHARACTER       UPLO
*      .. Local Arrays ..
real           A(LDA,NMAX)
*      .. External Subroutines ..
EXTERNAL        strtri, X04CAF
*      .. Executable Statements ..
WRITE (NOUT,*) 'F07TJF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN

*
*      Read A from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
      END IF

*
*      Compute inverse of A
*
      CALL strtri(UPLO,DIAG,N,A,LDA,INFO)

*
*      Print inverse
*
      WRITE (NOUT,*)
      IFAIL = 0
      CALL X04CAF(UPLO,DIAG,N,N,A,LDA,'Inverse',IFAIL)
END IF
STOP
*
END

```

## 9.2 Program Data

```
F07TJF Example Program Data
 4                               :Value of N
 'L'                             :Value of UPL0
 4.30
-3.96 -4.87
 0.40 0.31 -8.02
-0.27 0.07 -5.95 0.12 :End of matrix A
```

## 9.3 Program Results

F07TJF Example Program Results

```
Inverse
      1          2          3          4
1      0.2326
2     -0.1891    -0.2053
3      0.0043    -0.0079   -0.1247
4      0.8463    -0.2738   -6.1825    8.3333
```

---